*exemple extrait de l'aide de SAS sur la proc robustreg;
*DONNEES SIMULEES: un échantillon de 1000 observations dont les 900 sont issues d'un modèle 

linéaire y=10 + 5*x1 + 3*x2 + .5 * e 10  et les 100 autres ont été perturbées ce sont les points aberrants ( type point vertical)  ;
data a (drop=i);

   do i=1 to 1000;

      x1=rannor(1234);

      x2=rannor(1234);

      e=rannor(1234);

      if i > 900 then y=100 + e;

      else y=10 + 5*x1 + 3*x2 + .5 * e;

      output;

   end;

run;

ods graphics on;

* Régression linéaire classique: on voit dans les résultats qu'elle échoue à retrouver les vrais coefficients;
proc reg data=a;

   model y = x1 x2;

run;

*Régression robuste avec les différentes possibilités d'estimateurs robustes M-estimateurs, MM-estimateurs, 

S-estimateurs least trimmed sqaures: on voit qu'elles arrivent toutes à estimer correctement les coefficients du modèle;
proc robustreg data=a method=m plots=(rdplot ddplot histogram qqplot) ;

   model y = x1 x2 ;

run;

proc robustreg data=a method=mm seed=100;

   model y = x1 x2;

run;

proc robustreg data=a method=s seed=100;

   model y = x1 x2;

run;

proc robustreg data=a method=lts seed=100;

   model y = x1 x2;

run;

ods graphics off;

*AUTRE EXEMPLE autre type de données aberrantes;
data stack;

   input  x1 x2 x3 y exp$ @@;

   datalines;

80  27   89  42   e1   80  27   88  37   e2

75  25   90  37   e3   62  24   87  28   e4

62  22   87  18   e5   62  23   87  18   e6

62  24   93  19   e7   62  24   93  20   e8

58  23   87  15   e9   58  18   80  14  e10

58  18   89  14  e11   58  17   88  13  e12

58  18   82  11  e13   58  19   93  12  e14

50  18   89   8  e15   50  18   86   7  e16

50  19   72   8  e17   50  19   79   8  e18

50  20   80   9  e19   56  20   82  15  e20

70  20   91  15  e21

;

run;

*Régression robuste avec la méthode des M-estimateurs et diagnostics pour les points aberrants;
* voir dans le journal un warning signalant la présence de points aberrants pour lesquels la méthode LTS est plus adaptée;
proc robustreg data=stack;

   model y = x1 x2 x3 / diagnostics leverage;

   id    exp;

   run;

ods graphics on;

proc robustreg data=stack plots=(rdplot ddplot histogram qqplot);

   model y = x1 x2 x3;

run;

ods graphics off;

*Régression robuste avec la méthode LTS ;
proc robustreg data= stack fwls method=lts;

   model y = x1 x2 x3 / diagnostics leverage;

   id exp;

run;

